

Примери за потребителски дефинирани функции

База гани: SoftUni.sql

1) Employee Full Name by ID (Scalar)

```
DROP FUNCTION IF EXISTS dbo.ufn_GetEmployeeFullName;

GO

CREATE FUNCTION dbo.ufn_GetEmployeeFullName (@EmployeeId INT)
RETURNS NVARCHAR(200)
AS
BEGIN
    DECLARE @fullName NVARCHAR(200);

    SELECT @fullName =
        CONCAT(e.FirstName, ' ',
            COALESCE(NULLIF(e.MiddleName, '') + ' ', ''),
            e.LastName)
    FROM Employees AS e
    WHERE e.EmployeeID = @EmployeeId;

    RETURN @fullName;
END

GO
```

-- Example:

```
SELECT dbo.ufn_GetEmployeeFullName(1) AS FullName;
```

2) Manager Full Name for Employee (Scalar)

```
DROP FUNCTION IF EXISTS dbo.ufn_GetEmployeeManagerName;
```

```

GO

CREATE FUNCTION dbo.ufn_GetEmployeeManagerName (@EmployeeId INT)
RETURNS NVARCHAR(200)
AS
BEGIN
    DECLARE @managerName NVARCHAR(200);

    SELECT @managerName =
        CASE
            WHEN m.EmployeeID IS NULL THEN N'(no manager)'
            ELSE CONCAT(m.FirstName, ' ',
                COALESCE(NULLIF(m.MiddleName, '') + ' ', ''),
                m.LastName)
        END
    FROM Employees AS e
    LEFT JOIN Employees AS m ON m.EmployeeID = e.ManagerID
    WHERE e.EmployeeID = @EmployeeId;

    RETURN @managerName;
END
GO

```

-- Example:

```

SELECT dbo.ufn_GetEmployeeManagerName(5) AS ManagerName;

```

3) Employee Town by ID (Scalar)

```

DROP FUNCTION IF EXISTS dbo.ufn_GetEmployeeTown;

GO

```

```

CREATE FUNCTION dbo.ufn_GetEmployeeTown (@EmployeeId INT)
RETURNS VARCHAR(50)
AS
BEGIN
    DECLARE @town VARCHAR(50);

    SELECT @town = t.Name
    FROM Employees AS e
    JOIN Addresses AS a ON a.AddressID = e.AddressID
    JOIN Towns AS t ON t.TownID = a.TownID
    WHERE e.EmployeeID = @EmployeeId;

    RETURN @town;
END
GO

```

-- Example:

```

SELECT dbo.ufn_GetEmployeeTown(1) AS Town;

```

4) Department Employees Count by Department Name (Scalar)

```

DROP FUNCTION IF EXISTS dbo.ufn_DepartmentEmployeeCount;
GO

CREATE FUNCTION dbo.ufn_DepartmentEmployeeCount (@DepartmentName
VARCHAR(50))
RETURNS INT
AS
BEGIN
    DECLARE @cnt INT;

```

```

SELECT @cnt = COUNT(*)
FROM Departments AS d
JOIN Employees AS e ON e.DepartmentID = d.DepartmentID
WHERE d.Name = @DepartmentName;

RETURN ISNULL(@cnt, 0);

END

GO

-- Example:

SELECT dbo.ufn_DepartmentEmployeeCount('Engineering') AS EmployeesInDept;

```

5) Department Total Salary (Scalar)

```

DROP FUNCTION IF EXISTS dbo.ufn_DepartmentTotalSalary;

GO

CREATE FUNCTION dbo.ufn_DepartmentTotalSalary (@DepartmentId INT)
RETURNS MONEY
AS
BEGIN
    DECLARE @total MONEY;

    SELECT @total = SUM(e.Salary)
    FROM Employees AS e
    WHERE e.DepartmentID = @DepartmentId;

    RETURN ISNULL(@total, 0);

END

```

GO

-- Example:

```
SELECT dbo.ufn_DepartmentTotalSalary(1) AS TotalSalaryBudget;
```

6) Employees Count in Project (Scalar)

```
DROP FUNCTION IF EXISTS dbo.ufn_ProjectEmployeesCount;
```

GO

```
CREATE FUNCTION dbo.ufn_ProjectEmployeesCount (@ProjectId INT)
```

```
RETURNS INT
```

```
AS
```

```
BEGIN
```

```
    DECLARE @cnt INT;
```

```
    SELECT @cnt = COUNT(*)
```

```
    FROM EmployeesProjects AS ep
```

```
    WHERE ep.ProjectID = @ProjectId;
```

```
    RETURN ISNULL(@cnt, 0);
```

```
END
```

GO

-- Example:

```
SELECT dbo.ufn_ProjectEmployeesCount(3) AS PeopleOnProject;
```

7) Project Duration in Days by Project ID (Scalar)

(Uses GETDATE() when EndDate is NULL.)

```
DROP FUNCTION IF EXISTS dbo.ufn_ProjectDurationDays;
```

```
GO

CREATE FUNCTION dbo.ufn_ProjectDurationDays (@ProjectId INT)
RETURNS INT
AS
BEGIN
    DECLARE @start smalldatetime;
    DECLARE @end  smalldatetime;

    SELECT
        @start = p.StartDate,
        @end  = p.EndDate
    FROM Projects AS p
    WHERE p.ProjectID = @ProjectId;

    IF (@start IS NULL)
        RETURN NULL;

    IF (@end IS NULL)
        SET @end = GETDATE();

    RETURN DATEDIFF(DAY, @start, @end);
END

GO
```

-- Example:

```
SELECT dbo.ufn_ProjectDurationDays(1) AS DurationDays;
```

8) Employees in a Town (Inline Table-Valued Function)

```

DROP FUNCTION IF EXISTS dbo.udf_EmployeesByTown;

GO

CREATE FUNCTION dbo.udf_EmployeesByTown (@TownName VARCHAR(50))
RETURNS TABLE
AS
RETURN
(
    SELECT
        e.EmployeeID,
        CONCAT(e.FirstName, ' ', e.LastName) AS FullName,
        e.JobTitle,
        d.Name AS Department,
        t.Name AS Town
    FROM Employees AS e
    JOIN Departments AS d ON d.DepartmentID = e.DepartmentID
    JOIN Addresses AS a ON a.AddressID = e.AddressID
    JOIN Towns AS t ON t.TownID = a.TownID
    WHERE t.Name = @TownName
);

GO

```

-- Example:

```
SELECT * FROM dbo.udf_EmployeesByTown('Sofia');
```

9) Projects by Employee (Inline Table-Valued Function)

```

DROP FUNCTION IF EXISTS dbo.udf_ProjectsByEmployee;

GO

CREATE FUNCTION dbo.udf_ProjectsByEmployee (@EmployeeId INT)

```

RETURNS TABLE

AS

RETURN

(

SELECT

p.ProjectID,

p.Name AS ProjectName,

p.StartDate,

p.EndDate,

DATEDIFF(DAY, p.StartDate, ISNULL(p.EndDate, GETDATE())) AS DurationDays

FROM EmployeesProjects AS ep

JOIN Projects AS p ON p.ProjectID = ep.ProjectID

WHERE ep.EmployeeID = @EmployeeId

);

GO

-- Example:

SELECT * FROM dbo.udf_ProjectsByEmployee(7);

10) Top Paid Employees in Department (Multi-statement TVF)

DROP FUNCTION IF EXISTS dbo.udf_TopPaidEmployeesInDepartment;

GO

CREATE FUNCTION dbo.udf_TopPaidEmployeesInDepartment

(

@DepartmentName VARCHAR(50),

@TopN INT

)

RETURNS @result TABLE


```

(
    EmployeeID INT NOT NULL,
    FullName NVARCHAR(200) NOT NULL,
    Salary MONEY NOT NULL
)
AS
BEGIN
    INSERT INTO @result (EmployeeID, FullName, Salary)
    SELECT TOP (@TopN)
        e.EmployeeID,
        CONCAT(e.FirstName, ' ',
            COALESCE(NULLIF(e.MiddleName, '') + ' ', ''),
            e.LastName) AS FullName,
        e.Salary
    FROM Employees AS e
    JOIN Departments AS d ON d.DepartmentID = e.DepartmentID
    WHERE d.Name = @DepartmentName
    ORDER BY e.Salary DESC, e.EmployeeID ASC;

    RETURN;
END
GO

-- Example:
SELECT * FROM dbo.udf_TopPaidEmployeesInDepartment('Sales', 3);

```